Using PROC TABULATE® and ODS Style Options to Make Really Great Tables

Wendi L. Wright, CTB / McGraw-Hill, Harrisburg, PA

ABSTRACT

We start with an introduction to PROC TABULATE, looking at the basic syntax, and then building on this syntax by using examples. Examples will show how to produce one-, two-, and three-dimensional tables using the TABLE statement. More examples cover how to choose statistics for the table, labeling variables and statistics, how to add totals and subtotals, working with percents and missing data, and how to clean up the table using options. A look at the three ways you can use the PRELOADFMT option is also covered.

Next the ways to use the ODS STYLE= option in PROC TABULATE is covered. This option helps to customize the tables and improve their attractiveness. This option is very versatile and, depending on where the option is used, can justify cells or row and column headings, change colors for both the foreground and background of the table, modify lines and borders, controlling cell width, add a flyover text box in ODS HTML, or add GIF figures to the row or column headings.

INTRODUCTION

PROC TABULATE is a procedure that displays descriptive statistics in tabular format. It computes many statistics that other procedures compute, such as MEANS, FREQ, and REPORT and displays these statistics in a table format. TABULATE will produce tables in up to three dimensions and allows, within each dimension, multiple variables to be reported one after another hierarchically. There are also some very nice mechanisms that can be used to label and format the variables and the statistics produced.

BASIC SYNTAX

PROC TABULATE <options>;
  CLASS variables < / options>;
  VAR variables < / options>;
  TABLE <page> ,
    <row> ,
    column
    < / options> ;
... other statements ...
RUN;

Let's take a look at the basic syntax of the PROC TABULATE Procedure. We will start with three of the statements that you can use in PROC TABULATE, CLASS, VAR, and TABLE. As you can see each of these statements, as well as the PROC TABULATE statement itself allows options to be added. For each of the statements, the options need to be preceded with a ‘/’.

Note two differences in the syntax from any other Procedure in SAS®. One, the variables in all three statements can not be separated by commas. And two, the commas in the table statement are treated in a special way and mean a change in dimension.
VAR STATEMENT

The VAR statement is used to list the variables you intend to use to create summary statistics on. As such, they must be numeric.

CLASS STATEMENT

Classification variables allow you to get statistics by category. You will get one column or row for each value of the CLASS variable. You will need make sure you use a categorical variable with a limited number of categories or you may end up producing more pages of output than you intended.

The syntax for the CLASS statement is similar to the VAR statement. You list the variables you want to use for grouping data followed by a ‘/’ and any options you want. The variables here can be either numeric or character (unlike the VAR statement which requires numeric). The statistics you can request for these variables are only counts and percents. The statistics will be produced for each LEVEL of the variable. This is almost like using a BY statement within the table.

TABLE STATEMENT

The Table statement consists of up to three dimension expressions and the table options. To identify different dimensions, use commas. The order of the dimensions is page, row, and column. If you only specify one dimension, then it is assumed to be column. If two are specified, row, then column. Options appear at the end after a ‘/’.

You can have multiple table statements in one PROC TABULATE. This will generate one table for each statement. All variables listed in the table statement must also be listed in either the VAR or CLASS statements. In the table expressions, there are many statistics that can be specified. Among them are row and column percents, counts, means, and percentiles.

CONSTRUCTING A TABLE STATEMENT – DIMENSION EXPRESSIONS

There are many elements you can use to construct a table expression. You start with the variables you want to include in the table, but you can also specify the universal CLASS variable ALL which allows you to get totals. You will also need to specify what statistics you want to put in the cells of the table. To make your table ‘pretty’, you can also specify formats, labels, and ODS style specifications in the expression.

We begin by taking a closer look at constructing dimension expressions. Here is where PROC TABULATE differs from all the other Procedures in the SAS programming language. The syntax used for PROC TABULATE is very different.

- A comma specifies to add a new dimension.
- The asterisk is used to produce a cross tabulation of one variable with another (within the same dimension however, different from PROC FREQ).
- A blank is used to represent concatenation (i.e. place this output element after the preceding variable listed).
- Parenthesis will group elements and associate an operator with each element in the group
- Angle brackets specify a denominator definition for use in percentage calculations.
CONSTRUCTING A SINGLE DIMENSIONAL TABLE

The code below will produce a single dimension table. We have one VAR variable – income, and one CLASS variable – gender. To request what statistic you would like to see, use a ‘*’ and add the name of the statistic next to the VAR variable you want to use for producing the statistics. As demonstrated below, you can group multiple statistics and variables with parentheses to get the results you want. We are also requesting that mean income be produced for each level of the class variable gender.

Notice that there are NO commas included in the TABLES statement. This indicates to SAS that this is to be a one dimensional table.

```
PROC TABULATE data=one;
    CLASS GENDER;
    VAR income;
    TABLE income * (N Mean)  INCOME * MEAN * GENDER;
RUN;
```

<table>
<thead>
<tr>
<th>Income</th>
<th>Gender</th>
</tr>
</thead>
<tbody>
<tr>
<td></td>
<td></td>
</tr>
<tr>
<td>N</td>
<td>Mean</td>
</tr>
<tr>
<td>30.00</td>
<td>52111.80</td>
</tr>
<tr>
<td>Female</td>
<td>52000.69</td>
</tr>
<tr>
<td>Male</td>
<td>52238.79</td>
</tr>
</tbody>
</table>

ADDITION STATISTICS

The code above requested two statistics, N and Mean. You can produce many other statistics. Below is a table of them. If you do not provide a statistic name, the default statistic produced will be ‘N’ for the CLASS variables and ‘SUM’ for the VAR variables.

<table>
<thead>
<tr>
<th>Descriptive Statistics</th>
<th>Quantile Statistics</th>
</tr>
</thead>
<tbody>
<tr>
<td>COLPCTN</td>
<td>MEDIAN</td>
</tr>
<tr>
<td>PCTSUM</td>
<td>P1</td>
</tr>
<tr>
<td>COLPCTSUM</td>
<td>Q3</td>
</tr>
<tr>
<td>MAX</td>
<td>P90</td>
</tr>
<tr>
<td>ROWPCTN</td>
<td>P95</td>
</tr>
<tr>
<td>MEAN</td>
<td>P5</td>
</tr>
<tr>
<td>ROWPCTSUM</td>
<td>P10</td>
</tr>
<tr>
<td>MIN</td>
<td>P99</td>
</tr>
<tr>
<td>STDDEV / STD</td>
<td>Q1</td>
</tr>
<tr>
<td>N</td>
<td>RANGE</td>
</tr>
<tr>
<td>STDERR</td>
<td>Hypothesis Testing</td>
</tr>
<tr>
<td>NMISS</td>
<td>ProbT</td>
</tr>
<tr>
<td>SUM</td>
<td>T</td>
</tr>
<tr>
<td>PAGEPCTSUM</td>
<td></td>
</tr>
<tr>
<td>PCTN</td>
<td></td>
</tr>
<tr>
<td>VAR</td>
<td></td>
</tr>
</tbody>
</table>
TWO DIMENSIONAL TABLE

To create a two dimensional table, we need to add some additional detail to our TABLES statement. We specify first the row dimension, then the column dimension, separated by a comma. You can get very different table structures by changing where the statistic definitions are placed. They can be attached to either the VAR or the CLASS variable, but the numbers in the cells will ALWAYS be calculated using the VAR variable(s).

Here we have the statistics attached to the columns.

```
PROC TABULATE data=one;
   CLASS gender;
   VAR income;
   TABLE gender,
       income * (N Mean Max) ;
RUN;
```

<p>| | | |</p>
<table>
<thead>
<tr>
<th></th>
<th></th>
<th></th>
</tr>
</thead>
<tbody>
<tr>
<td>Gender</td>
<td></td>
<td></td>
</tr>
<tr>
<td>Female</td>
<td>16.00</td>
<td>52000.69</td>
</tr>
<tr>
<td>Male</td>
<td>14.00</td>
<td>52238.79</td>
</tr>
</tbody>
</table>

If you move the statistic specification so that it is attached to the rows, the results look very different.

```
PROC TABULATE data=one;
   CLASS gender;
   VAR income;
   TABLE gender * (N Mean Max) ,
       income ;
RUN;
```

<p>| | | |</p>
<table>
<thead>
<tr>
<th></th>
<th></th>
<th></th>
</tr>
</thead>
<tbody>
<tr>
<td>Gender</td>
<td></td>
<td></td>
</tr>
<tr>
<td>Female</td>
<td>N</td>
<td>16.00</td>
</tr>
<tr>
<td></td>
<td>Mean</td>
<td>52000.69</td>
</tr>
<tr>
<td></td>
<td>Max</td>
<td>93849.00</td>
</tr>
<tr>
<td>Male</td>
<td>N</td>
<td>14.00</td>
</tr>
<tr>
<td></td>
<td>Mean</td>
<td>52238.79</td>
</tr>
<tr>
<td></td>
<td>Max</td>
<td>78695.00</td>
</tr>
</tbody>
</table>
MORE THAN ONE CLASSIFICATION VARIABLE

You can specify multiple classification variables. They can be used in any of the dimensions and can be nested. In our example, we have three CLASS variables. Two of the CLASS variables are nested in the row dimension, Fulltime * Gender.

When you have multiple CLASS variables, I recommend using the option MISSING. By default, The CLASS statement tells SAS to drop observations that have a missing value in even ONE of the variables. If you specify the MISSING option in your CLASS statement, SAS does NOT remove that observation from all the tables and will consider the missing values as valid levels for the CLASS variable(s).

```sas
PROC TABULATE data=one;
  CLASS gender fulltime educ / MISSING;
  VAR income;
  TABLE fulltime * gender ,
    Income * educ * mean ;
RUN;
```

<table>
<thead>
<tr>
<th>Gender</th>
<th>Educ</th>
<th>Income</th>
</tr>
</thead>
<tbody>
<tr>
<td></td>
<td>High School</td>
<td>Bachelors</td>
</tr>
<tr>
<td></td>
<td>Mean</td>
<td>Mean</td>
</tr>
<tr>
<td>Fulltime</td>
<td>Male</td>
<td>52506.67</td>
</tr>
<tr>
<td>Female</td>
<td>42771.20</td>
<td>74589.60</td>
</tr>
<tr>
<td>Male</td>
<td>44788.67</td>
<td>36947.00</td>
</tr>
<tr>
<td>Parttime</td>
<td>Female</td>
<td>35600.00</td>
</tr>
</tbody>
</table>


ADDING TOTALS AND SUBTOTALS

In order to get marginal statistics in your table, you use the ‘ALL’ keyword. You can use the keyword in multiple places and, depending on where you put the keyword, there will be different subtotals produced. You can place the keyword on the row or the column dimensions or on both. In this example we are requesting subtotals for Gender and overall totals for both the row and the column dimensions.

```sas
PROC TABULATE data=one;
   CLASS gender fulltime educ;
   TABLE (fulltime ALL) * gender ALL,
         educ * N;
RUN;
```

<table>
<thead>
<tr>
<th></th>
<th>Educ</th>
</tr>
</thead>
<tbody>
<tr>
<td></td>
<td>High School</td>
</tr>
<tr>
<td></td>
<td>N</td>
</tr>
<tr>
<td>Fulltime</td>
<td>Gender</td>
</tr>
<tr>
<td>Fulltime</td>
<td>Female</td>
</tr>
<tr>
<td></td>
<td>Male</td>
</tr>
<tr>
<td>Parttime</td>
<td>Female</td>
</tr>
<tr>
<td></td>
<td>Male</td>
</tr>
<tr>
<td>All</td>
<td>Female</td>
</tr>
<tr>
<td></td>
<td>Male</td>
</tr>
<tr>
<td>All</td>
<td></td>
</tr>
</tbody>
</table>

A few additional code examples are here. These demonstrate that the ALL keyword can be used as many times as you wish.

```sas
PROC TABULATE data=one;
   CLASS gender fulltime educ;
   TABLE fulltime * (gender ALL) ,
         (educ all)* N ;
RUN;

PROC TABULATE data=one;
   CLASS gender fulltime educ;
   TABLE (fulltime ALL) * (gender ALL) * ALL ,
         (educ all)* N ;
RUN;
```
ADDING AND HIDING ROW AND COLUMN LABELS

There are two ways to add labels for your variables. The first is the simplest. Just add the text ‘=label’ after the variable name to the TABLE dimension expression. This will work for both variables and statistics. The second way is to add a LABEL statement for variables and/or a KEYLABEL statement for statistics to your code. For example:

```sas
LABEL var='label';
KEYLABEL stat='label';
```

In order to hide variable or statistic labels, you leave the label specification blank (i.e. ‘ ’).

The following example demonstrates the simpler way to add and hide labels.

```sas
PROC TABULATE data=one;
   CLASS educ gender fulltime;
   VAR income;
   TABLE educ ,
       Fulltime='Employment Status' * gender = ' ' *
           income * mean = ' ';
RUN;
```

<table>
<thead>
<tr>
<th>Educ</th>
<th>Employment Status</th>
<th>Fulltime</th>
<th>Parttime</th>
</tr>
</thead>
<tbody>
<tr>
<td>High School</td>
<td>52506.67</td>
<td>44788.67</td>
<td>35600.00</td>
</tr>
<tr>
<td>Bachelors</td>
<td>42771.20</td>
<td>36947.00</td>
<td></td>
</tr>
<tr>
<td>Masters</td>
<td>50729.25</td>
<td>62258.00</td>
<td></td>
</tr>
<tr>
<td>Doctorate</td>
<td>74589.60</td>
<td>56466.00</td>
<td></td>
</tr>
</tbody>
</table>

Alternatively, you can use this code to produce the table above:

```sas
PROC TABULATE data=one;
   CLASS educ gender fulltime;
   VAR income;
   TABLE educ,
       Fulltime * gender * income * mean ;
   LABEL gender=' ' Fulltime='Employment Status';
   KEYLABEL mean=' ';
RUN;
```
FILLING THE BIG WHITE BOX

To fill in the big white box in the upper left, use the BOX= option.

```
PROC TABULATE data=one;
  CLASS gender fulltime;
  VAR income;
  TABLE fulltime = 'Employment Status',
               Gender * income * mean
               / BOX='Mean Income';
RUN;
```

<table>
<thead>
<tr>
<th>Mean Income</th>
<th>Gender</th>
</tr>
</thead>
<tbody>
<tr>
<td></td>
<td>Female</td>
</tr>
<tr>
<td>Income</td>
<td></td>
</tr>
<tr>
<td>Mean</td>
<td></td>
</tr>
</tbody>
</table>

<table>
<thead>
<tr>
<th>Employment Status</th>
<th>Gender</th>
</tr>
</thead>
<tbody>
<tr>
<td>Fulltime</td>
<td>58680.40</td>
</tr>
<tr>
<td>Parttime</td>
<td>40867.83</td>
</tr>
</tbody>
</table>

THREE DIMENSIONAL TABLES

Three dimensional tables are easy to produce, just add another section BEFORE the row and column expressions in the table statement. PROC TABULATE will now interpret the dimension statements in this order, first page, then the row, then the columns.

Three dimensional tables have a nice way to fill in the upper left area. Instead of the label of the page dimension appearing above the table, you can use the BOX=_page_ option to place that label inside the big white box. Only a part of the output is included here.

```
PROC TABULATE data=one;
  CLASS gender fulltime educ;
  VAR income;
  TABLE educ='Education',
          fulltime = 'Employment Status',
               Gender * income * mean
               / BOX=_PAGE_;
RUN;
```

This will produce a separate table for each level of education with the value of the variable education in the big white box.
PRODUCING CELL PERCENTS

One of the nice things about PROC TABULATE is that it allows you to put both percents and summary statistics in the same table. You can get percentages on both numeric and character variables as the percents are based only on counts. The VAR statement is not needed unless you want to add other summary statistics besides count and percents. Just use the CLASS statement. Several types of percents can be requested or you can construct your own percentages by specifying the denominator you wish to use. The use of a complex denominator will not be covered here. We will only cover the three percents that are most commonly used:

- PCTN - percent of total.
- ROWPCTN – percent across row (use row total as denominator).
- COLPCTN – percent across column (use column total as denominator).

This example shows the use of ROWPCTN. Note that the percents will add up to 100% (taking into account rounding errors) across each row. COLPCTN works similarly – the columns will add up to 100%.

```sas
PROC TABULATE data=one;
  CLASS ethnic educ;
  TABLE ethnic * ROWPCTN, Educ;
RUN;
```

<table>
<thead>
<tr>
<th>Ethnic</th>
<th>High School</th>
<th>Bachelors</th>
<th>Masters</th>
<th>Doctorate</th>
</tr>
</thead>
<tbody>
<tr>
<td>Asian</td>
<td>25.00</td>
<td>75.00</td>
<td></td>
<td></td>
</tr>
<tr>
<td>Af.Amer.</td>
<td>33.33</td>
<td>16.67</td>
<td>33.33</td>
<td>16.67</td>
</tr>
<tr>
<td>Hisp.</td>
<td>28.57</td>
<td>14.29</td>
<td>28.57</td>
<td>28.57</td>
</tr>
<tr>
<td>Am. Ind.</td>
<td>50.00</td>
<td></td>
<td>25.00</td>
<td>25.00</td>
</tr>
<tr>
<td>White</td>
<td>27.27</td>
<td>27.27</td>
<td>27.27</td>
<td>18.18</td>
</tr>
</tbody>
</table>
**SPECIAL THINGS YOU CAN DO WITH FORMATS – THE MLF OPTION**

Normally when you create a format, you cannot have repeated values on the left side of the format definition. The Multilabel option in PROC FORMAT allows you to specify the same value to apply to different groups. To apply this in a PROC TABULATE, you must use the MLF (multi-label format) option on the CLASS statement for the variable you want multiple groups for. Note that the sort order will be alpha by the formatted values. If you want other orders, use the ORDER= option. Shown here is the ORDER=DATA option with a presort of the data by education level.

```sas
proc format;
  value $educmlf (multilabel)
    '1' = 'High School'
    '2','3','4'='College'
    '2' = 'Bachelors'
    '3' = 'Masters'
    '4' = 'Doctorate'
  ;
run;

proc sort data=one;
  by educ;
run;

proc tabulate data=one;
  class educ / mlf order=data;
  var income;
  tables educ='Education Level',
          Income='Income'*(N Mean);
  format educ $educmlf.;
run;
```

<table>
<thead>
<tr>
<th>Education Level</th>
<th>N</th>
<th>Mean</th>
</tr>
</thead>
<tbody>
<tr>
<td>High School</td>
<td>8</td>
<td>45385.75</td>
</tr>
<tr>
<td>College</td>
<td>23</td>
<td>55663.83</td>
</tr>
<tr>
<td>Bachelors</td>
<td>9</td>
<td>44966.33</td>
</tr>
<tr>
<td>Masters</td>
<td>7</td>
<td>55670.14</td>
</tr>
<tr>
<td>Doctorate</td>
<td>7</td>
<td>69411.43</td>
</tr>
</tbody>
</table>

**SPECIAL THINGS YOU CAN DO WITH FORMATS – THE PRELOADFMT OPTION**

The PRELOADFMT option can be used in three ways depending on what other options you use with it. This option has no effect unless you use one of three other options with it AND you also assign formats to the class variable(s).

- Use with the PRINTMISS option to see include all categories from the format in the table regardless of whether data exists for any individual category.
- Use with the EXCLUSIVE option to include ONLY those categories from the format in the table even if data exists for other values.
- Use with ORDER=DATA if you want PROC TABULATE to use the order the levels in the table in the same order they appear in the PROC FORMAT. With this option, you will also need to use the (NOTSORTED) option in the PROC FORMAT VALUE statement.
Example 1: Use PRELOADFMT with the PRINTMISS option:

```sas
proc format;
   value $ethnic
       'W'='White'
       'H'='Hisp.'
       'I'='Am. Ind.'
       'A'='Asian'
       'B'='Af.Amer.'
       'L'='Alaskan Native'
       ' '='Missing'
;
proc tabulate data=one;
   class ethnic / PRELOADFMT;
   var income;
   tables ethnic,
       Income*(N Mean)
       / PRINTMISS;
   format ethnic $ethnic.;
run;
```

<table>
<thead>
<tr>
<th>Ethnic</th>
<th>N</th>
<th>Mean</th>
</tr>
</thead>
<tbody>
<tr>
<td>Asian</td>
<td>2</td>
<td>53474.00</td>
</tr>
<tr>
<td>Af.Amer.</td>
<td>5</td>
<td>58866.00</td>
</tr>
<tr>
<td>Hisp.</td>
<td>5</td>
<td>46563.40</td>
</tr>
<tr>
<td>Am. Ind.</td>
<td>4</td>
<td>62382.50</td>
</tr>
<tr>
<td>Alaskan Native</td>
<td>0</td>
<td>.</td>
</tr>
<tr>
<td>White</td>
<td>11</td>
<td>47694.27</td>
</tr>
</tbody>
</table>

Example 2: Use PRELOADFMT with the EXCLUSIVE option:

```sas
proc format;
   value $ethEXCL
       'H'='Hisp.'
       'I'='Am. Ind.'
       'L'='Alaskan Native'
       ' '='Missing'
;
proc tabulate data=one;
   class ethnic / PRELOADFMT EXCLUSIVE;
   var income;
   tables ethnic,
       Income*(N Mean);
   format ethnic $ethEXCL.;
run;
```

<table>
<thead>
<tr>
<th>Ethnic</th>
<th>N</th>
<th>Mean</th>
</tr>
</thead>
<tbody>
<tr>
<td>Hisp.</td>
<td>5</td>
<td>46563.40</td>
</tr>
<tr>
<td>Am. Ind.</td>
<td>4</td>
<td>62382.50</td>
</tr>
</tbody>
</table>

Example 3: Use PRELOADFMT with the DATA=ORDER option:

```sas
proc format;
   value $ethnic (notsorted)
       'W'='White'
       'H'='Hisp.'
       'I'='Am. Ind.'
       'A'='Asian'
       'B'='Af.Amer.'
       'L'='Alaskan Native'
       ' '='Missing'
;
proc tabulate data=one order=data;
   class ethnic / preloadfmt;
   var income;
   tables ethnic,
       Income*(N Mean);
   format ethnic $ethnic.;
run;
```

<table>
<thead>
<tr>
<th>Ethnic</th>
<th>N</th>
<th>Mean</th>
</tr>
</thead>
<tbody>
<tr>
<td>White</td>
<td>11</td>
<td>47694.27</td>
</tr>
<tr>
<td>Hisp.</td>
<td>5</td>
<td>46563.40</td>
</tr>
<tr>
<td>Am. Ind.</td>
<td>4</td>
<td>62382.50</td>
</tr>
<tr>
<td>Asian</td>
<td>2</td>
<td>53474.00</td>
</tr>
<tr>
<td>Af.Amer.</td>
<td>5</td>
<td>58866.00</td>
</tr>
</tbody>
</table>
ADDING FORMATS TO THE STATISTICS AND REMOVING HORIZONTAL SEPARATORS

Let's take a look at two simple options to improve the table's look. The first option allows you to specify formats for the numbers in the cells of the table using the *F=fmt. expression. The second option is the NOSEPS option which removes the horizontal dividers between the row values from your table.

```sas
PROC TABULATE data=one NOSEPS;
  CLASS educ gender fulltime / missing;
  VAR income;
  TABLE educ=' ' * (fulltime=' ' ALL),
              gender=' ' * Income=' ' * ( N*F=6. MEAN*F=Dollar8. );
RUN;
```

<table>
<thead>
<tr>
<th></th>
<th>Female</th>
<th></th>
<th>Male</th>
<th></th>
</tr>
</thead>
<tbody>
<tr>
<td></td>
<td>N</td>
<td>Mean</td>
<td>N</td>
<td>Mean</td>
</tr>
<tr>
<td>High School</td>
<td>N</td>
<td>Mean</td>
<td>N</td>
<td>Mean</td>
</tr>
<tr>
<td>Fulltime</td>
<td>3</td>
<td>$44,789</td>
<td>3</td>
<td>$52,507</td>
</tr>
<tr>
<td>Parttime</td>
<td>2</td>
<td>$35,600</td>
<td>5</td>
<td>$45,744</td>
</tr>
<tr>
<td>All</td>
<td>5</td>
<td>$42,771</td>
<td>0</td>
<td>0</td>
</tr>
<tr>
<td>Bachelors</td>
<td>3</td>
<td>$36,947</td>
<td>0</td>
<td>0</td>
</tr>
<tr>
<td>Parttime</td>
<td>7</td>
<td>$40,587</td>
<td>0</td>
<td>0</td>
</tr>
<tr>
<td>All</td>
<td>12</td>
<td>$50,729</td>
<td>0</td>
<td>0</td>
</tr>
<tr>
<td>Masters</td>
<td>4</td>
<td>$50,729</td>
<td>0</td>
<td>0</td>
</tr>
<tr>
<td>Parttime</td>
<td>3</td>
<td>$62,258</td>
<td>0</td>
<td>0</td>
</tr>
<tr>
<td>All</td>
<td>7</td>
<td>$55,670</td>
<td>0</td>
<td>0</td>
</tr>
<tr>
<td>Doctorate</td>
<td>0</td>
<td>$74,590</td>
<td>2</td>
<td>$56,466</td>
</tr>
<tr>
<td>All</td>
<td>5</td>
<td>$74,590</td>
<td>2</td>
<td>$56,466</td>
</tr>
</tbody>
</table>

CLEAN UP ROW HEADERS WITH INDENT AND RTS OPTIONS

Two more options to improve the look of your table are INDENT= which is very nice for subsetting row subheaders and RTS= which specifies how wide you want the row header field to be. Note, the RTS= count includes the | bar dividers at the beginning and the end of the row header fields, so include these in your count.

```sas
PROC TABULATE data=one NOSEPS;
  CLASS educ gender fulltime / missing;
  VAR income;
  TABLE educ=' ' * (fulltime=' ' ALL),
              gender=' ' * Income=' ' * ( N*F=6. MEAN*F=Dollar8. )
              / BOX='Income' INDENT=3 RTS=12;
RUN;
```

<table>
<thead>
<tr>
<th>Income</th>
<th>Female</th>
<th></th>
<th>Male</th>
<th></th>
</tr>
</thead>
<tbody>
<tr>
<td></td>
<td>N</td>
<td>Mean</td>
<td>N</td>
<td>Mean</td>
</tr>
<tr>
<td>High School</td>
<td>N</td>
<td>Mean</td>
<td>N</td>
<td>Mean</td>
</tr>
<tr>
<td>Fulltime</td>
<td>3</td>
<td>$44,789</td>
<td>3</td>
<td>$52,507</td>
</tr>
<tr>
<td>Parttime</td>
<td>2</td>
<td>$35,600</td>
<td>5</td>
<td>$45,744</td>
</tr>
<tr>
<td>All</td>
<td>5</td>
<td>$42,771</td>
<td>0</td>
<td>0</td>
</tr>
<tr>
<td>Bachelors</td>
<td>3</td>
<td>$36,947</td>
<td>0</td>
<td>0</td>
</tr>
<tr>
<td>Parttime</td>
<td>7</td>
<td>$40,587</td>
<td>0</td>
<td>0</td>
</tr>
<tr>
<td>All</td>
<td>12</td>
<td>$50,729</td>
<td>0</td>
<td>0</td>
</tr>
<tr>
<td>Masters</td>
<td>4</td>
<td>$50,729</td>
<td>0</td>
<td>0</td>
</tr>
<tr>
<td>Parttime</td>
<td>3</td>
<td>$62,258</td>
<td>0</td>
<td>0</td>
</tr>
<tr>
<td>All</td>
<td>7</td>
<td>$55,670</td>
<td>0</td>
<td>0</td>
</tr>
<tr>
<td>Doctorate</td>
<td>0</td>
<td>$74,590</td>
<td>2</td>
<td>$56,466</td>
</tr>
<tr>
<td>All</td>
<td>5</td>
<td>$74,590</td>
<td>2</td>
<td>$56,466</td>
</tr>
</tbody>
</table>
ODS STYLE ELEMENTS

ODS Style elements can also be used to change the look of a table. A few are listed below. These elements all work with HTML, PDF, RTF, PS and PCL destinations. Other elements are listed in the “SAS 9.1 Output Delivery System: User’s Guide” that you can find at http://support.sas.com/documentation/ onlinedoc/91pdf/.

<table>
<thead>
<tr>
<th>Style Element</th>
<th>Description</th>
</tr>
</thead>
<tbody>
<tr>
<td>Foreground/Background=</td>
<td>modify color</td>
</tr>
<tr>
<td>BorderWidth=</td>
<td>specify thickness of borders</td>
</tr>
<tr>
<td>Just/Vjust=</td>
<td>specify horizontal and vertical justification</td>
</tr>
<tr>
<td>Font_Face/Font_Weight/Font_Size=</td>
<td>change font characteristics</td>
</tr>
<tr>
<td>Rules=</td>
<td>specify horizontal and vertical rule dividers</td>
</tr>
<tr>
<td>CellWidth/CellHeight=</td>
<td>change size of table cells</td>
</tr>
<tr>
<td>Cellpadding/CellSpacing=</td>
<td>specify white space and thickness of spacing around cell</td>
</tr>
<tr>
<td>OutputWidth=</td>
<td>specify width of table</td>
</tr>
</tbody>
</table>

WHERE DO YOU PUT THE STYLE ELEMENTS?

Depending on where you place the style options, many different results can be achieved. If you place the style options on the PROC TABULATE statement, for example, you will affect all the table cells. See below for a list of some of the different places where you can put the style options and what portion of the table they will affect.

Note that for the CLASS, CLASSLEV, VAR, and KEYWORD statements, the style options can also be specified in the dimension expression in the Table statement.

<table>
<thead>
<tr>
<th>Style Place In</th>
<th>Part of Table Affected</th>
</tr>
</thead>
<tbody>
<tr>
<td>PROC TABULATE S=[ …]</td>
<td>data cells</td>
</tr>
<tr>
<td>CLASS varname / S=[ …]</td>
<td>heading for variable varname</td>
</tr>
<tr>
<td>CLASSLEV varname / S=[ …]</td>
<td>class values for variable varname</td>
</tr>
<tr>
<td>VAR varname / S=[ …]</td>
<td>heading for variable varname</td>
</tr>
<tr>
<td>KEYWORD stat / S=[ …]</td>
<td>heading for named stat</td>
</tr>
<tr>
<td>TABLE page,row,col / S=[ …]</td>
<td>table borders, rules, cell spacing</td>
</tr>
<tr>
<td>BOX={label=' ' S=[ …] }</td>
<td>table Box</td>
</tr>
</tbody>
</table>

CHANGING THE FOREGROUND (TEXT) OR BACKGROUND COLOR

Here is an example showing how to change the color in various parts of the table. Notice that we are applying a black foreground (text) to all cells in the table. For the gender variable we are assigning a background of yellow to the values only (not to the column header) using the CLASSLEV statement. For the mean statistic label we are using a foreground of white and a background of purple.
ODS RTF file='c:\myfile.rtf';
PROC TABULATE data=one f=10.2 S=[foreground=black];
   CLASS gender;
   CLASSLEV gender / S=[background=yellow];
   VAR income;
   TABLE gender=' ' all={label='Tot'},
               mean={s=[foreground=white background=purple]} * income
       / box={label='Income'};
Run;
ODS RTF close;

<table>
<thead>
<tr>
<th></th>
<th>Mean</th>
</tr>
</thead>
<tbody>
<tr>
<td></td>
<td>Income</td>
</tr>
<tr>
<td>Female</td>
<td>52000.69</td>
</tr>
<tr>
<td>Male</td>
<td>54089.53</td>
</tr>
<tr>
<td>Tot</td>
<td>53011.42</td>
</tr>
</tbody>
</table>

**ADDING JUSTIFICATION**

This example shows how to add justification to three sections of the table. We are going to center all the cells inside the table, right justify the total 'Tot' row label, and Bottom Right justify the text inside the upper left box.

You can use ODS Style options to add many things to the table. Here is an example that adds color and justification. We are setting the foreground (print type) to be black and all the cells should be centered. Note the use of the new CLASSLEV statement that we have not seen before.

ODS RTF file='c:\myfile.rtf';
PROC TABULATE data=one f=10.2 S=[foreground=black just=c];
   CLASS gender;
   CLASSLEV gender / S=[background=yellow];
   VAR income;
   TABLE gender=' ' all={label='Tot'},
               mean={s=[foreground=white background=purple]} * income
       / box={label='Income' s=[VJust=B Just=R]};
Run;
ODS RTF close;

<table>
<thead>
<tr>
<th></th>
<th>Mean</th>
</tr>
</thead>
<tbody>
<tr>
<td></td>
<td>Income</td>
</tr>
<tr>
<td>Female</td>
<td>52000.69</td>
</tr>
<tr>
<td>Male</td>
<td>54089.53</td>
</tr>
<tr>
<td>Tot</td>
<td>53011.42</td>
</tr>
</tbody>
</table>
CHANGING CELL WIDTH

This example shows how to change the cell width of both the row headers and the columns of the table.

```sas
ODS RTF file='c:\myfile.rtf';
PROC TABULATE data=one f=10.2 S=[foreground=black just=c cellwidth=200];
   CLASS gender / S=[cellwidth=250];
   CLASSLEV gender / S=[background=yellow];
   VAR income;
   TABLE gender=' ' all={label='Tot' s=[just=R]},
       mean={s=[foreground=white background=purple]} * income
       box={label='Income' s=[VJust=B Just=R]};
RUN;
ODS RTF close;
```

<table>
<thead>
<tr>
<th></th>
<th>Mean</th>
</tr>
</thead>
<tbody>
<tr>
<td></td>
<td>Income</td>
</tr>
<tr>
<td>Female</td>
<td>52000.69</td>
</tr>
<tr>
<td>Male</td>
<td>54089.53</td>
</tr>
<tr>
<td>Tot</td>
<td>53011.42</td>
</tr>
</tbody>
</table>

REMOVING THE LINES IN THE TABLE AND ADJUSTING CELL SPACING

With the addition of the rules, cellspacing and cellpadding options we can remove all lines from the table.

```sas
ODS RTF file='c:\myfile.rtf';
PROC TABULATE data=one f=10.2 S=[foreground=black cellwidth=200 just=c];
   CLASS gender;
   CLASSLEV gender / S=[background=yellow];
   VAR income;
   TABLE gender=' ' all={label='Tot' s=[just=R]},
       mean={s=[foreground=white background=purple]} * income
       box={label='Income' s=[VJust=B Just=R]};
RUN;
ODS RTF close;
```

<table>
<thead>
<tr>
<th></th>
<th>Mean</th>
</tr>
</thead>
<tbody>
<tr>
<td></td>
<td>Income</td>
</tr>
<tr>
<td>Female</td>
<td>52000.69</td>
</tr>
<tr>
<td>Male</td>
<td>54089.53</td>
</tr>
<tr>
<td>Tot</td>
<td>53011.42</td>
</tr>
</tbody>
</table>
ADDING A FLYOVER TEXT BOX USING ODS AND PROC FORMAT

You can use formats with styles to do a number of tricks. One thing you can do is to add a flyover text box to your row or column header cells. The result can not be demonstrated here, but what happens is when you hold your curser over the column header when you view the table in a Web Browser, a little text box will open underneath that shows the text you have in your format.

PROC FORMAT:
VALUE $ethnic
  'W'='White - Non-Hispanic'
  'H'='Hispanic American'
  'I'='American Indian'
  'A'='Asian'
  'B'='African American'
  '.'='Missing'
;
ODS HTML file='c:\myfile.HTML';
PROC TABULATE data=one;
CLASS ethnic gender;
CLASSLEV ethnic / s=[flyover=$ethnic.];
VAR income;
TABLE gender,
  ethnic * income;
RUN;
ODS HTML CLOSE;

USING PROC FORMAT AND ODS TO ADD PICTURES TO YOUR TABLE

Another trick is to add a figure to the row or column header cell. Note that in this example, the gif files needed to be in the same directory that the HTML file was in.

PROC FORMAT:
VALUE $gendGif
  'M'='bluebutterfly.gif'
  'F'='pinkbutterfly.gif';
RUN;
ODS HTML file='c:\myfile.HTML';
PROC TABULATE data=one;
CLASS GENDER;
CLASSLEV gender / S=[VJust=T postimage=$gendGif. Cellwidth=80];
VAR income;
TABLE income * (N Mean)
  INCOME * MEAN * GENDER;
RUN;
ODS HTML CLOSE;
The next trick I am going to mention is how to use a format to highlight different cells with different colors based on the final value in the cell.

```
Title 'Highlighting Cell Values with Colors';
ODS HTML FILE='c:\myfile.html';
PROC FORMAT;
   VALUE watchit
       0 - 20000 = 'Green'
       20001 - 40000 = 'Orange'
       40001 - 50000 = 'Blue'
       50001 - 60000 = 'Purple'
       60001 - high = 'Red';
RUN;
PROC TABULATE data=one S=[foreground=watchit.];
   CLASS gender ethnic / MISSING;
   VAR income;
   TABLE gender ALL,
       Income * (ethnic ALL) * Mean;
RUN;
ODS HTML CLOSE;
```

### ADDING COLORS TO CELLS BASED ON VALUES IN THE CELLS

The next trick I am going to mention is how to use a format to highlight different cells with different colors based on the final value in the cell.

```
Title 'Highlighting Cell Values with Colors';
ODS HTML FILE='c:\myfile.html';
PROC FORMAT;
   VALUE watchit
       0 - 20000 = 'Green'
       20001 - 40000 = 'Orange'
       40001 - 50000 = 'Blue'
       50001 - 60000 = 'Purple'
       60001 - high = 'Red';
RUN;
PROC TABULATE data=one S=[foreground=watchit.];
   CLASS gender ethnic / MISSING;
   VAR income;
   TABLE gender ALL,
       Income * (ethnic ALL) * Mean;
RUN;
ODS HTML CLOSE;
```
ADDING COLORS TO ROW/COLUMN HEADERS BASED ON THEIR VALUES

The last trick is how to change the background color of the row headers based on the value of the class variable.

```sas
Title 'Highlighting Row Headers with Colors';
ODS HTML FILE='c:\myfile.html';

PROC FORMAT;
    VALUE $gendbac
       'M'='blue'
       'F'='pink';
    VALUE $gendfor
       'M'='white'
       'F'='black';
RUN;

PROC TABULATE data=one;
    CLASS gender;
    CLASSLEV gender /
        S=[background=$gendbac. Foreground=$gendfor. Cellwidth=100];
    VAR income;
    TABLE gender ALL,
        Income * (N Mean Std);
RUN;

ODS HTML CLOSE;
```

<table>
<thead>
<tr>
<th>Gender</th>
<th>N</th>
<th>Mean</th>
<th>Std</th>
</tr>
</thead>
<tbody>
<tr>
<td>Female</td>
<td>16</td>
<td>52000.69</td>
<td>23861.47</td>
</tr>
<tr>
<td>Male</td>
<td>15</td>
<td>54089.53</td>
<td>20034.90</td>
</tr>
<tr>
<td>All</td>
<td>31</td>
<td>53011.42</td>
<td>21751.54</td>
</tr>
</tbody>
</table>
CONCLUSION

Proc Tabulate can greatly increase your ability to produce informative, beautiful, and professional looking tables. This paper covered many of the ways to produce a great table, but the options described here are by no means exhaustive of what you can do with PROC TABULATE.

There are several very good resources that you can use to learn more about PROC TABULATE. Among them are:

- PROC TABULATE by Example, by Lauren Haworth
- SAS Guide to Report Writing: Examples, by Michele M. Burlew
- SAS Guide to TABULATE Processing

AUTHOR CONTACT

Your comments and questions are valued and welcome. Contact the author at:

Wendi L. Wright
1351 Fishing Creek Valley Rd.
Harrisburg, PA 17112
Phone: (717) 513-0027
E-mail: wendi_wright@ctb.com

SAS and all other SAS Institute Inc. products or service names are registered trademarks or trademarks of SAS Institute, Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.